LAG PLOT

PURPOSE

Generates a lag plot.

DESCRIPTION

For a single time series, a lag plot is a graphical data analysis technique for determining if an autocorrelation structure exists within the time series. For two time series, a lag plot is a graphical technique for determining if cross-correlation structure exists between the two time series. Ideally (for a white noise time series or for 2 uncorrelated time series), the lag plot should have the appearance of a random shotgun pattern. Any kind of a structured pattern in a lag plot indicates an underlying auto/cross-correlation model, the nature of which may be inferred from the type of lag plot structure.

In time series analysis, a lag is a fixed time displacement. For example, y(2) and y(7) would be said to have a lag of 5 (= 7-2). For a lag plot, the lag is fixed at some value specified by the analyst. The default value is a lag of 1.

For a lag plot on a single time series, the lag plot consists of:

Vertical axis = x(i)
Horizontal axis = x(i+lag)

For a lag plot for 2 time series, the lag plot consists of:

Vertical axis = y(i)
Horizontal axis = x(i+lag)

SYNTAX 1

LAG <n> PLOT <x> <SUBSET/EXCEPT/FOR qualification>

where <n> is an integer number or parameter between 1 and n-1 (n is the number of observations) that specifies the lag;
<x> is the variable of raw data values which is being analyzed for autocorrelation structure;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for a single time series.

SYNTAX 2

LAG <n> PLOT <y1> <y2> <SUBSET/EXCEPT/FOR qualification>

where <n> is an integer number or parameter between 1 and n-1 (n is the number of observations) that specifies the lag;
<y1> is the first variable of raw data values which is being analyzed for cross-correlation structure;
<y2> is the second variable of raw data values which is being analyzed for cross-correlation structure;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for two time series.

EXAMPLES

```
LAG 3 PLOT X
LAG PLOT X
LAG -12 PLOT Y X
LAG PLOT Y X
```

DEFAULT

If <n> is omitted, the default lag is 1.

SYNONYMS

None

RELATED COMMANDS

CHARACTERS	=	Sets the types for plot characters.
LINES	=	Sets the types for plot lines.
PLOT	=	Generates a data or function plot.
4-PLOT	=	Generates a 4-plot for univariate analysis.
CORRELATION PLOT	=	Generates an auto or cross-correlation plot.
SPECTRUM	=	Generates a spectral plot.
SUMMARY	=	Generates a table of summary statistics.

LET	=	Computes various statistics (and many other capabilities).
FIT	=	Carries out a least squares fit.

APPLICATIONS

Time Series Analysis, Regression

IMPLEMENTATION DATE

Pre-1987

PROGRAM 1

```
. THIS SAMPLE PROGRAM READS THE FILE LEW.DAT IN THE DATAPLOT
. REFERENCE DIRECTORY. THE DATA IS BEAM DELECTION DATA.
SKIP 25
READ LEW.DAT Y
.
LEGEND 1 AUTOCORRELATION ANALYSIS
LEGEND 2 LAG PLOT
TITLE AUTOMATIC
Y1LABEL X(LC()i)
CHAR X
LINES
YMAX 500
XMAX 500
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
LOOP FOR K = 1 1 4
    X1LABEL X(LC()I+^K)
    LAG ^K PLOT Y
END OF LOOP
END OF MULTIPLOT
```



PROGRAM 2

```
. THIS SAMPLE PROGRAM READS THE FILE HAYES1.DAT IN THE DATAPLOT
. REFERENCE DIRECTORY. THIS IS FIRE RESEARCH SMOKE OBSCURATION DATA.
.
SKIP 25
READ HAYES1.DAT JUNK Y1 Y2
.
TITLE AUTOMATIC
LEGEND 1 CROSS-CORRELATION ANALYSIS
LEGEND 2 LAG PLOT
Y1LABEL X(LC()i)
TIC OFFSET 0.2 0.2
.
CHAR X
LINES
YLIMITS 0 3
XLIMITS 0 3
MULTIPLOT 2 2
MULTIPLOT CORNER COORDINATES 0 0 100 100
LOOP FOR K = 1 1 4
      X1LABEL Y(LC()I+^K)
      LAG ^K PLOT Y1 Y2
END OF LOOP
END OF MULTIPLOT
```